
Preamble

The new NetRexx language. which has been designed by M.F.Cowlishaw as the successor of the classic REXX language, has some important differences in notation and semantics to the REXX language.

Whilst REXX essentially is a very powerful procedural scripting language, with natural structured statements, NetRexx is a blend of  REXX and Java, and fully object-oriented.

Unfortunately, the two languages are NOT upwards compatible: Major language differences include:

· REXX stems (ie. Abc.def) have changed their notation and have been replaced by indexed strings (i.e. abc[def]), and need to be defined before their first usage.

· NetRexx introduces the capability/necessity to declare variables in front of their usage.

· expressions on instruction level are not automatically thrown as COMMANDS to the underlying language command processor.

· labels are not allowed in NetRexx, at least not on top of the following code.

· the ‘SIGNAL label’ statement is not available in NetRexx, and may no longer be used as a  simple ‘GO TO’ label equivalent.

· procedure calls (function invokations of typeless(void) functions) do not use the verb CALL, and parameters of a  procedure call must be enclosed in parenthesis in NetRexx.

· variables, indexed strings (stems) and labels must all have distinct names (in REXX, variables with the same name as a stem or function are allowed)

· Literal strings may include escape sequences (started by a \) in NetRexx, and thus REXX literals must be translated, if they contain a backslash.

· The notation for invoking the imbedded (builtin)  functions (methods) is object oriented in NetRexx, and thus all references to builtin functions need to be changed

· The concept of external functions has been replaced by the concept of packages, and thus all references to external functions need to be changed.

· The handling of the builtin stack (verbs PULL, PUSH,QUEUE, and the QUEUED() function in the REXX language) is not supported in NetRexx.

· A couple of standard Rexx functions are not available in NetRexx (like DATE, TIME, ...) and thus need to be translated to the equivalent Java routines when possible, or an appropriate emulation must be provided.

· Although NetRexx is able to execute NetRexx code immediately (without the need to compile JAVA code) the verb INTERPRET and the standard REXX  functions SYMBOL(…) and VALUE(…)   are no longer available.

· The concept of interrupt handling is different.

· The DO loop versions of the DO instruction have been  replaced by the LOOP instruction.

· The concept of LABELS to head a sequence of statements has been removed, or replaced by new syntactic equivalents (i.e. DO LABEL label ..., LOOP LABEL label ... …)

· Whilst the PARSE instruction is still available (thank you, Mike!) it’s semantics have slightly been changed and do no longer support qualified variables (i.e. abc[def]) in the template body.

· Formal arguments are no longer defined by the ARG and PARSE ARG instructions, and thus multiple ARG or PARSE ARG instructions must be resolved by proper methods.

· etc, etc.

Hence, whilst the two languages look very similiar at the first glance (as they have very similiar structured statements available), the language differences in syntax and semantics are significant. Thus, it is not possible to directly use programs written in REXX  within NetRexx, as a significant amount of recoding is needed. 

The Rexx to NetRexx converter (Rexx2Nrx) has been developped to support you in the task to convert existing REXX programs or program parts to NetRexx. This converter parses your classic REXX program, analyses it, and finally converts it to NetRexx as far as possible. 

As an immediate advantage, your classic REXX programs become available to all JAVA platforms, and don’t  need any platform specific code anymore.

Installation

The converter is delivered as a single ZIP-file containing a number of JAVA classes, their corresponding NetRexx source code,  and some external control tables which drive the converter. 

It is recommended that you put the converter in a separate directory,\Rexx2Nrx\, for instance.

Unpack the Rexx2Nrx.ZIP file as usual using the EXTRACT command of WINZIP. Your \Rexx2Nrx\  directory should now contain the following files:

Rexx2Nrx.doc

The documentation of the converter (The document you are just 

reading)

Rexx2Nrx.bat    
A short batch file invoking the REXX to NetRexx converter, with detailed conversion messages

Rx2n.bat
the short form of Rexx2Nrx, without detailed conversion messages, and with implicit NetRexx and Java compilation, too. 

Rexx.Builtin


The REXX builtin function definition file

Rexx2Nrx.renames
A file containing the necessary RENAME rules (as your classic REXX code  might use some JAVA reserved words as variable names/labels).

Please list this file and adopt it to your needs as necessary.

Rexx2Nrx.jar
A jar-file containing the run-time package routines, mainly emulations for the classic Rexx runtime-routines not provided by NetRexx. This JAR-file also contains the classes ‘rex_pars’ – the parser and ‘rex_nrx’  - the actual  Rexx to NetRexx converter, as well as some utility routines delivered with the package.

Readme.txt

the read-me file, which contains some basic instructions as well as a list of known bugs/open issues.

  The run-time package provided includes:

RexxStk.class
implements the classic REXX Stacking mechanisms (the classic REXX verbs pull, push, queue  and the queued() function).

RexxTime.class 


implements the classic REXX functions Date and Time.

RexxFile.class
implements the File-IO routines stream, linein, lineout, lines and chars, charin, charout as defined for classic Rexx, but with an object oriented fashion. Whilst the original versions of the converter did use Max Marsigliettis RXFile, the whole package has now been re-written from scratch for performance reasons  and compatibility to CMS or Risc 6000, for instance.

RexxBits.class
The Rexx Bits class and associated methods (namely bitand, bitor, and bitxor). These routines are currenlty NOT yet released!

SysCmd.class
A number of methods to execute System commands, as well as Rexx,  NetRexx, or Java commands.

What to do next

In order to work with the Rexx2Nrx converter, you now have to do 2 steps:

1.)  Copy or move the Rexx2Nrx run time package file (Rexx2Nrx.jar) to your JAVA runtime extensions directory, typically called ‚\java\jre\lib\ext‘ , so that NetRexx and JAVA will find the JAVA classes of the converter and the run-time. Altenatively, you may modify your CLASSPATH to directly point to \Rexx2Nrx\Rexx2Nrx.jar.

2.)  Modify your  PATH variable to include \Rexx2Nrx\, as the converter will need to find

a couple of batch-files and auxiliary files distributed with the package. These files include:

Rexx2Nrx.bat, Rx2n.bat, Rexx.Builtin, Rexx2Nrx.renames.

Otherwise you will have to copy those files to every directory where you are executing

the converter from!


Invoking the converter

The classic REXX to NetRexx converter is usually invoked from the shell with the command 

Rexx2Nrx (or rx2n for brevity)

cmd>Rexx2Nrx abc


(for the verbose form with detailed conversion messages)  

cmd>rx2n abc                       (for the short form without detailed conversion messages, and  

  with implicit NetRexx and Java compilation)      

 
where abc.rexx  (or abc.exec or abc.cmd)  contains your source ‘classic REXX’  program: Note that the filetype (extension) must be ommited, as it is provided automatically. Otherwise the NetRexx Compiler would be invoked with your source Rexx program. Hence, please OMIT the filetype when invoking rx2n.

The file-types rexx, rex, cmd, exec, srx, rexxincl, or rexxpack are searched for the corresponding source file (in this sequence).

Options might be entered in UNIX/Windows style, prefixed by a hyphen,

cmd>Rx2n modulename –binary –quiet –notime -nostats

     or may be entered in CMS-style (enclosed in parenthesis),  for example

cmd>Rexx2Nrx modulename (binary quiet verbose1)

Conversion steps

The conversion is carried out in the following steps:

1.) First, the given  file is parsed, and prepared for final conversion.

2.) Second, the actual conversion takes plasce.

Here is a simple example (thanks to M.F.Colishaw for this nice piece of Rexx code)

Example 1: The QTSMALL program of M.F. Cowlishaw, author of Rexx and NetRexx
/*----------------------------------------------------------*/

/* QT.  This program displays the time in real English.     */

/* Two argument strings may be supplied.  If "?" is given   */

/* as the first argument then the program displays a        */

/* description of itself.  If a second argument is supplied */

/* it is used as a test value to check the operation of the */

/* program.  This second value must be a time in the format */

/* HH:MM:SS, and does not have its syntax checked.          */

/*----------------------------------------------------------*/

/*--------- First process the argument strings -------------*/

parse arg parm testtime .      /* get the argument strings */

select

  when parm='?' then call tell            /* say what we do */

  when parm=''  then nop          /* OK (no first argument) */

  when parm='-' then nop  /* ===> OK, maybe second argument */

  otherwise

    say 'The only valid parameter to QT is "?".  The argument'

    say 'that you supplied ("'parm'") has been ignored.'

    call tell    /* usually helpful to describe the program */

  end

if testtime='' then now=time()    /* default - use time now */

               else now=testtime  /* caller gave test value */

/*--------- Now start processing in earnest ----------------*/

/* Nearness phrases - using a compound variable as example  */

near.0=''                                          /* exact */

near.1=' just gone';  near.2=' just after'         /* after */

near.3=' nearly';     near.4=' almost'            /* before */

/* Extract the hours, minutes, and seconds from the time.   */

parse var now hour':'min':'sec

if sec>29 then min=min+1                /* round up minutes */

mod=min//5              /* where we are in 5 minute bracket */

out="It's"near.mod             /* start building the result */

if min>32 then hour=hour+1         /* we are TO the hour... */

min=min+2     /* shift minutes to straddle a 5-minute point */

/* Now special-case the result for Noon and Midnight hours  */

if hour//12=0 & min//60<=4 then do

  if hour=12 then say out 'Noon.'

             else say out 'Midnight.'

  exit                              /* we are finished here */

  end

min=min-(min//5)                     /* find nearest 5 mins */

if hour>12

 then hour=hour-12              /* get rid of 24-hour clock */

 else

  if hour=0 then hour=12       /* .. and allow for midnight */

/* Determine the phrase to use for each 5-minute segment    */

select

  when min= 0 then nop               /* add "o'clock" later */

  when min=60 then min=0                           /* ditto */

  when min= 5 then out=out 'five past'

  when min=10 then out=out 'ten past'

  when min=15 then out=out 'a quarter past'

  when min=20 then out=out 'twenty past'

  when min=25 then out=out 'twenty-five past'

  when min=30 then out=out 'half past'

  when min=35 then out=out 'twenty-five to'

  when min=40 then out=out 'twenty to'

  when min=45 then out=out 'a quarter to'

  when min=50 then out=out 'ten to'

  when min=55 then out=out 'five to'

  end

numbers='one two three four five six',    /* (continuation) */

  'seven eight nine ten eleven twelve'

out=out word(numbers,hour)           /* add the hour number */

if min=0 then out=out "o'clock"  /* .. and o'clock if exact */

say out'.'                      /* display the final result */

exit

/*----------------------------------------------------------*/

/* Subroutine that describes the purpose of the program     */

/*----------------------------------------------------------*/

Tell:

 say 'QT will display the current time in real English.'

 say 'Call without any arguments to display the time, or with'

 say '"?" to display this information.  A second argument (in'

 say 'the format "HH:MM:SS") will be used as a test value.'

 say 'British English idioms are used in this program.'

 say /* blank line - we are about to continue and show time */

 return

/* Mike Cowlishaw,  December 1979 - January 1985            */

 
When the classic REXX to NetRexx converter is invoked, a progress report is displayed on the screen. Concurrently, these displays are recorded in the log-file ‘PP.LOG’ for further inspection. PP.LOG is the general preprocessor log-file used by the converter.

For the example above, the LOG-file looks as follows:

File: qtsmall1.rexx

===================

classic REXX Parsing of file: qtsmall1.rexx with rex_pars.nrx vs 2.22

Copyright (c) Thomas Schneider, 2000, all rights reserved.

  label: Tell

List of defined functions/subroutines:

======================================

  1. Internal Nothing  Subroutine tell

  2. Builtin  Text     Function   time

  3. Builtin  Any      Function   word

List of all Labels used:

========================

 No Label                                    : defined in file/at line

--- ---------------------------------------- : -----------------------

  1 Internal Subroutine Tell                 : qtsmall1.rexx line: 83

      169 code-instructions saved in: qtsmall1.$code

       92 lines of file: qtsmall1.rexx parsed.

classic REXX Parsing used      0.94  seconds (   0.010 sec. by line)

================================================================================

Performing necessary renames to avoid Java/NetRexx conflicts

============================================================

       59 rename-rules loaded from: Rexx2Nrx.renames

File: qtsmall1

==============

NetRexx translation of file: qtsmall1 with rex_nrx.nrx vs. 2.22

Copyright (c) Thomas Schneider, 2000, all rights reserved.

      131 item-declarations read from: qtsmall1.$decl

      287 code-instructions loaded from: qtsmall1.$code

        1 labels defined

Code analysis in logical order of execution

===========================================

analysing: main program ( line: 1-90 )

   analysing: Subroutine tell ( line: 83-90 )

      Internal Subroutine tell returns Nothing

   Internal Subroutine tell returns Nothing

checking Stems & simple variables for duplicate names:

======================================================

note, that in NetRexx simple Variables and subscripted

strings (stems in Rexx) must have different names! 

       no names changed due to name-conflicts

rename local variables when needed/wanted:

==========================================

note, that properties in NetRexx might NOT have the

same name as a local name or argument of a method !

Convert backslash, imbedded quotes, and operators

=================================================

converting back_slashes

converting imbedded quotes

converting operators to NetRexx standards

Converting declarations when necessary

======================================

Declaration Conversions used:      0.031 seconds.

Generating NetRexx module: qtsmall1

===================================

# 1 translating: Tell() returns Nothing

Rexx to NetRexx converter used:     0.72 seconds ( 0.01198 sec/line )

The conversion report above has been produced with the standard verbosity (verbose3),
You may tailor this report to your needs by the verbosen option (see below).


The generated final NetRexx program is put to the associated NRX-file, and will look as follows, when you did use options type, for example (see below):

/*--------------------------------------------------------------------*/

/* QT.  This program displays the time in real English.               */

/* Two argument strings may be supplied.  If "?" is given             */

/* as the first argument then the program displays a                  */

/* description of itself.  If a second argument is supplied           */

/* it is used as a test value to check the operation of the           */

/* program.  This second value must be a time in the format           */

/* HH:MM:SS, and does not have its syntax checked.                    */

/*--------------------------------------------------------------------*/

/*                                                                    */

/*--------- First process the argument strings -------------          */

/*====================================================================*/

/* generated by the classic REXX to NetRexx converter                 */

/*                     (c) Thomas.Schneider@Donauland.at              */

/*====================================================================*/

/* orig. REXX source: qtsmall1                                        */

/*      generated at: 2001-11-03 19:04:15                             */

/*              with: rex_nrx.nrx vs. 2.22                            */

/*  Rexx2Nrx options: type                                            */

/*  IO-method used  : 3 ( RexxFile Object oriented )                  */

/*====================================================================*/

import Rexx2Nrx.Rexx2RT.RexxTime                                        

class qtsmall1                                                          

properties public static;                                               

/*====================================================================*/

   /* ... Declare & init global REXX variables first:                 */

   /* ... Declare FileDescriptors ( derived from : FileName )         */

   /* ... Declare Rexx Identifiers ( Id : Id )                        */

   /* ... Declare Flags ( Var : Boolean Logical Compare )             */

   /* ... Declare Counters ( Var : Number Byte Short Int Long )       */

   hour = Int 0                              /* we are TO the hour... */

   min = Int 0                                    /* round up minutes */

   sec = Int 0                                                          

   mod = Int 0                    /* where we are in 5 minute bracket */

   /* ... Declare Decimals ( Var : Decimal Numeric Float Double )     */

   /* ... Declare Strings ( Var : Char String Rexx Text )             */

   parm = Rexx " "                                                      

   testtime = Rexx ''                                                   

   now = Rexx ''                            /* default - use time now */

   out = Rexx ''                         /* start building the result */

   numbers = Rexx ''                                /* (continuation) */

   /* ... Declare Stems / Arrays ( Stem S-Var Array Q-Var )           */

   near = Rexx ''                                                       

   /* ... Declare Other Global Variables                              */

   /* ... End of common REXX variable declarations                    */

/*====================================================================*/

method main(args=String[]) static;                                      

arg=Rexx(args) -- program arguments as single string                    

arg=arg -- avoid NetRexx warning                                        

   parse arg parm testtime .              /* get the argument strings */

   select;                                                              

      when parm = '?' then do;                      /* say what we do */

         tell()                                                         

         end /*when*/                       /* OK (no first argument) */

      when parm = '' then do;                                           

         end /*when*/               /* ===> OK, maybe second argument */

      when parm = '-' then do;                                          

         end /*when*/                                                   

      otherwise do;                                                     

         say 'The only valid parameter to QT is "?".  The argument'     

         say 'that you supplied ("'||parm||'") has been ignored.'       

         tell()            /* usually helpful to describe the program */

      end; /* otherwise */                                              

   end; /* select */                                                    

   if testtime = '' then do;                /* default - use time now */

      now = RexxTime.time();                                            

   end                                      /* caller gave test value */

   else do;                                                             

      now = testtime;                                                   

/*--------- Now start processing in earnest ----------------          */

/* Nearness phrases - using a compound variable as example            */

   end; /*if*/                                               /* exact */

   near[0] = '';                                                        

   near[1] = ' just gone';                                   /* after */

   near[2] = ' just after';                                             

   near[3] = ' nearly';                                     /* before */

   near[4] = ' almost';                                                 

/* Extract the hours, minutes, and seconds from the time.             */

   parse now hour ':' min ':' sec                                       

   if sec > 29 then do;                           /* round up minutes */

      min = min + 1;                                                    

   end; /*if*/                    /* where we are in 5 minute bracket */

   mod = min // 5;                                                      

   out = "It's"||near[mod];              /* start building the result */

   if min > 32 then do;                      /* we are TO the hour... */

      hour = hour + 1;                                                  

   end; /*if*/          /* shift minutes to straddle a 5-minute point */

   min = min + 2;                                                       

/* Now special-case the result for Noon and Midnight hours            */

   if ((hour // 12) = 0) & ((min // 60) <= 4) then do;                  

      if hour = 12 then do;                                             

         say out 'Noon.'                                                

      end                                                               

      else do;                                                          

         say out 'Midnight.'                                            

      end; /*if*/                             /* we are finished here */

      exit ;                                                            

   end; /*if*/                                 /* find nearest 5 mins */

   min = min - (min // 5);                                              

   if hour > 12 then do;                                                

      hour = hour - 12;                   /* get rid of 24-hour clock */

   end                                                                  

   else do;                                                             

      if hour = 0 then do;               /* .. and allow for midnight */

         hour = 12;                                                     

/* Determine the phrase to use for each 5-minute segment              */

      end; /*if*/                                                       

   end; /*if*/                                                          

   select;                                                              

      when min = 0 then do;                    /* add "o'clock" later */

         end /*when*/                                        /* ditto */

      when min = 60 then do;                                            

         min = 0;                                                       

         end /*when*/                                                   

      when min = 5 then do;                                             

         out = out 'five past';                                         

         end /*when*/                                                   

      when min = 10 then do;                                            

         out = out 'ten past';                                          

         end /*when*/                                                   

      when min = 15 then do;                                            

         out = out 'a quarter past';                                    

         end /*when*/                                                   

      when min = 20 then do;                                            

         out = out 'twenty past';                                       

         end /*when*/                                                   

      when min = 25 then do;                                            

         out = out 'twenty-five past';                                  

         end /*when*/                                                   

      when min = 30 then do;                                            

         out = out 'half past';                                         

         end /*when*/                                                   

      when min = 35 then do;                                            

         out = out 'twenty-five to';                                    

         end /*when*/                                                   

      when min = 40 then do;                                            

         out = out 'twenty to';                                         

         end /*when*/                                                   

      when min = 45 then do;                                            

         out = out 'a quarter to';                                      

         end /*when*/                                                   

      when min = 50 then do;                                            

         out = out 'ten to';                                            

         end /*when*/                                                   

      when min = 55 then do;                                            

         out = out 'five to';                                           

      end;                                                              

   end; /* select */                                                    

   numbers = 'one two three four five six' 'seven eight nine ten'-

      'eleven twelve';                              /* (continuation) */

   out = out numbers.word(hour);               /* add the hour number */

   if min = 0 then do;                     /* .. and o'clock if exact */

      out = out "o'clock";                                              

   end; /*if*/                            /* display the final result */

   say out||'.'                                                         

   exit ;                                                               

/*--------------------------------------------------------------------*/

/* Subroutine that describes the purpose of the program               */

/*--------------------------------------------------------------------*/

method Tell() static public ;                                           

   say 'QT will display the current time in real English.'              

   say 'Call without any arguments to display the time, or with'        

   say '"?" to display this information.  A second argument (in'        

   say 'the format "HH:MM:SS") will be used as a test value.'           

   say 'British English idioms are used in this program.'               

   say         /* blank line - we are about to continue and show time */

   return ;                                                             

/* Mike Cowlishaw,  December 1979 - January 1985                      */

end of example 1: QTSMALL1.

A couple of notes are worthwhile:

1. The converter performs an ‘automatic typing’ algorithm trying to find the ‘best type’ for each item. In order to do so, the source code is analysed in ‘logical order of execution’, i.e. each local subroutine/function/procedure is analysed at it’s first potential point of execution. When you use option type, the item-type detected is used to specify the type of each item. As a default, option NOTYPE is used, however, and the default  typing og NetRexx apllies.
2. For each local subroutine, function, or procedure, a static method is generated implementing the same functionality.

3. All global variables are defined ahead at the top of the program unit, with their proper NetRexx Type, and a proper initialization value. Note that you may change the proper  types used by options –short (for short integers) and options –binary (see below). Also note, that by default each yet unused variable is initialized to the UPPERCASE variable name by default in classic Rexx, a feature which is NOT available in NetRexx. Those variables are called an ‘Id’ in the converter.

4. When necessary, local variables are renamed so that they don’t conflict with global variables or function parameters.

5. When necessary, simple variables are renamed so that they don’t conflict with a  Rexx stem of the same name (Note, that in classic Rexx a stem may have the same name as a simple variable, which is NOT allowed in NetRexx …)

6. The same renaming applies, if function names and variable names conflict (note again, that in classic Rexx a function and a variable might have the same name)

7. To avoid conflicts with Java, all variables with a name identical to a Java reserved word are renamed, in turn. These renames may be tailored by the entries in file ‘Rexx2Nrx.renames’, which is provided in a standard form for your usage and/or expansion.

8. All string literals are analysed and converted to the proper NetRexx syntax when necessary. In particular, as the backslash is an escape-character in NetRexx, all single backslashes are replicated to conform to NetRexx syntax.

9. All references to qualified variables (stems in classic Rexx) are changed to their proper NetRexx syntax, i.e:   ‘abc.def.hig’ becomes  ‘abc[def,hig]’

10. All references to builtin methods (like word, wordpos, etc) become changed in their notation to use the NetRexx object oriented style. Those builtin methods are defined in file Rexx.Builtin, and maybe tailored by you to your specific needs when appropriate.

11. When classic REXX builtin functions or verbs are used, which don’t have an equivalent in NetRexx, than an appropriate emulation is called, e.g.:

  

QUEUE abc         becomes RexxStk.
Queue(abc)
                        PULL abc            becomes RexxStk.pull_upper(abc)
Those methods are provided in the Rexx2Nrx.jar file, which is deployed with the package..     

12. Last, not least, the generated code is indented properly to reformat your program nicely for readability. This has, however, the trade-off that the line-numbers will no longer refer to your original Rexx program. Note, however, that the converter is usually invoked only for the conversion process, as many users will like to tune the converted program after the initial conversion effort.

13. A complete, annotated listing of all items comprising your program and the translations executed is available in the $LIST-file for your inspection, if option LIST is used.

14. Example 1: $LIST-file :

Example 2:

The second example uses the short form of the classic Rexx to NetRexx converter,  and shows some Important design changes to get appropriate performance when possible: 

/* The corresponding OS/2 REXX procedure of

   the example2.nrx (see that).

   Note: The pure REXX version is 5.5 

   times faster on my system:

   486dx4-120, 32meg RAM, 2xFireball 1.2g HD. */

/***********************************************************/

/* cannot recall where I got this example from             */

/* timing added, Th. Schneider, 01.08.2001                 */

/* note that RexxFile.charin() reads a bunch af chars      */

/* at once to get efficient IO !                           */

/***********************************************************/

 say ' '

 say 'This program creates a copy of an existing'

 say 'binary file.'

 say 'The result file of the copy is placed in'

 say 'this directory, under the name TEMP.BIN.'

 say ' '

 say 'WARNING! This example takes forever to execute'

 say 'with large files..'

 do until length(fname) \= 0

   say ' '

   say 'Please enter a valid file name..'

   say ' '

   say 'Example: c:\filename.exe'

   say ' '

   ok=charout(,"File name>")

   fname = linein()

   if length(stream(fname,"c", "query exists")) = 0 then

   do

     fname = ""

     say "That file doesn't exists."

   end

 end

 "@del temp.bin"

 time1=time('R')  /* reset timer */

 ok = stream(fname,"c","open read")

 ok = stream("temp.bin","c","open write")

 count = 0

 do while chars(fname) > 0

  bytes = charin(fname)

  /*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>*/

  /* here is the difference: in RexxFile, the method     */

  /* charin(file) reads by default a bunch (chunk)     */

  /* of characters, not a single character, at once !! */

  /* the differenes in run-time are significant      !!*/

  /* The Rexx to NetRexx converter will give an appropriate */

  /* attention-message to warn the user                   !!*/ 

  /*       Th. Schneider, 02.11.2001 */

  /*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>*/

  count = count + length(bytes)

  remaining = charout("temp.bin", bytes)

  if remaining > 0 then do

    say 'cannot copy:' count'.th byte:' bytes

    exit 99

  end  

 end

 say count 'bytes copied.'

 time2=time('E') /* get elapsed time */

 elapsed= format(time2-time1,4,3) /* get elapsed time */ 

 say 'copying used:' elapsed 'seconds.'  

The generated program looks as follows:

/* The corresponding OS/2 REXX procedure of                           */

/*   the example2.nrx (see that).                                     */

/*   Note: The pure REXX version is 5.5                               */

/*   times faster on my system:                                       */

/*   486dx4-120, 32meg RAM, 2xFireball 1.2g HD.                       */

/*                                                                    */

/**********************************************************************/

/* cannot recall where I got this example from                        */

/* timing added, Th. Schneider, 01.08.2001                            */

/* note that RexxFile.charin() reads a bunch af chars                 */

/* at once to get efficient IO !                                      */

/**********************************************************************/

/*                                                                    */

/*====================================================================*/

/* generated by the classic REXX to NetRexx converter                 */

/*                     (c) Thomas.Schneider@Donauland.at              */

/*====================================================================*/

/* orig. REXX source: example2.cmd                                    */

/*      generated at: 2001-11-03 19:22:35                             */

/*              with: rex_nrx.nrx vs. 2.22                            */

/*  Rexx2Nrx options: verbose0 nologo nolist nostats notime           */

/*  IO-method used  : 3 ( RexxFile Object oriented )                  */

/*====================================================================*/

import Rexx2Nrx.Rexx2RT.SysCmd                                          

import Rexx2Nrx.Rexx2RT.RexxFile                                        

import Rexx2Nrx.Rexx2RT.RexxTime                                        

class example2 uses RexxFile                                            

properties public static;                                               

/*====================================================================*/

   /* ... Declare & init global REXX variables first:                 */

   /* ... Declare FileDescriptors ( derived from : FileName )         */

   FD_fname = RexxFile Null                                             

   FD_36 = RexxFile("temp.bin")                                         

   /* ... Declare Rexx Identifiers ( Id : Id )                        */

   /* ... Declare Flags ( Var : Boolean Logical Compare )             */

   /* ... Declare Counters ( Var : Number Byte Short Int Long )       */

   time1 = 0                                           /* reset timer */

   count = 0                                                            

   remaining = 0                                                        

   time2 = 0                                      /* get elapsed time */

   /* ... Declare Decimals ( Var : Decimal Numeric Float Double )     */

   /* ... Declare Strings ( Var : Char String Rexx Text )             */

   ok = ''                                                              

   bytes = " "                                                          

   elapsed = ''                                   /* get elapsed time */

   /* ... Declare Stems / Arrays ( Stem S-Var Array Q-Var )           */

   /* ... Declare Other Global Variables                              */

   fname = ''                                                           

   /* ... End of common REXX variable declarations                    */

/*====================================================================*/

method main(args=String[]) static;                                      

arg=Rexx(args) -- program arguments as single string                    

arg=arg -- avoid NetRexx warning                                        

   say ' '                                                              

   say 'This program creates a copy of an existing'                     

   say 'binary file.'                                                   

   say 'The result file of the copy is placed in'                       

   say 'this directory, under the name TEMP.BIN.'                       

   say ' '                                                              

   say 'WARNING! This example takes forever to execute'                 

   say 'with large files..'                                             

   loop until fname.length() \= 0                                       

      say ' '                                                           

      say 'Please enter a valid file name..'                            

      say ' '                                                           

      say 'Example: c:\\filename.exe'                                   

      say ' '                                                           

      ok = RexxFile.FD().charout("File name>");                         

      fname = RexxFile.FD().linein();                                   

      FD_fname = RexxFile.FD(fname)                                     

      if FD_fname.FileFullName.length() = 0 then do;                    

         fname = "";                                                    

         FD_fname = RexxFile.FD(fname)                                  

         say "That file doesn't exists."                                

      end; /*if*/                                                       

   end; /* loop  */                                                     

   RexxFile.FD("temp.bin").delete() /* == '@del'... */                  

   time1 = RexxTime.time('R');                         /* reset timer */

   ok = FD_fname.open("read");                                          

   ok = FD_36.open("write");                                            

   count = 0;                                                           

   loop while FD_fname.chars() > 0                                      

      bytes = FD_fname.charin();                                        

/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>   */

/* here is the difference: in RexxFile, the method                    */

/* charin(file) reads by default a bunch (chunk)                      */

/* of characters, not a single character, at once !!                  */

/* the differenes in run-time are significant      !!                 */

/* The Rexx to NetRexx converter will give an appropriate             */

/* attention-message to warn the user                   !!            */

/*       Th. Schneider, 02.11.2001                                    */

/*>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>  */

      count = count + bytes.length();                                   

      remaining = FD_36.charout(bytes);                                 

      if remaining > 0 then do;                                         

         say 'cannot copy:' count||'.th byte:' bytes                    

         exit 99;                                                       

      end; /*if*/                                                       

   end; /* loop  */                                                     

   say count 'bytes copied.'                                            

   time2 = RexxTime.time('E');                    /* get elapsed time */

   elapsed = (time2 - time1).format(4,3);         /* get elapsed time */

   say 'copying used:' elapsed 'seconds.'                               

a couple of notes are worthwhile again:

1.)  when builtin functions of classic Rexx are used, those are translated to proper Netrexx emulations. Appropriate import-statements are inserted at the top of the program. The stream-function, for example, belongs to  Rexx2Nrx.Rexx2RT.RexxFile. You will also recognize, that proper object oriented methods are used for File I/O, for instance. See RexxFile.doc for a detailed documentation of those routines.

2.) commands, like the ‘@del temp.bin’ in the example above, are  thrown to the SysCmd.cmd routine. A standard version of SysCmd is provided, which emulates some

CMS functions like ‘MAKEBUF’ or ‘DESBUF’, for instance. Note, howver,  that system commands which are literally notated, are directly converted to the proper Java method when possible, thus eliminating the overhead of a system command.

When you like to support commands from other operating systems, you should enhance

the SysCmd routine. (the NetRexx source code is provided as a basis ). For the example above, the  OS/2 ‘@DEL’ command to delete a file has been added.  

3.) For performance reasons, the ‘charin’ function uses a default ‘length’ of 8K bytes, so

that 8K bytes are  read at once by default to speed things up a bit…

4.) rx2n, the short form of the converter, automatically invokes the NetRexx & Java compiler after successful conversion for you convenience. 

Thus, example2 may be directly executed as usual:

cmd>java example2

This program creates a copy of an existing

binary file.

The result file of the copy is placed in

this directory, under the name TEMP.BIN.

WARNING! This example takes forever to execute

with large files..

Please enter a valid file name..

Example: c:\filename.exe

File name>rex_nrx.nrx

325381 bytes copied.

copying used:    3.969 seconds.



Operation modes :


The operation of the converter may be adapted to your needs by a number of options, which  should work as close as possible to the equivalent NetRexx options. These options include :

 

-nologo
suppress the LOGO information at the top of  each converter 


step













-
-notime
suppress the timing messages given



-nostats
suppress all statistic informations



-type

explicitely type all generated Rexx variables 

                        
-verbosen 
where ‘n’ is a proper value between 0 and 5 :




set the ‘verbosity’ of the converter output.




0 
is the lowest level,  and will only display warnings, 


errors, and attention messages.



           3 
is the default level 




5 
is the highest level (do use with care, might give a lot of  


auto-debugger messages)


-pause

pause after giving a full screen of messages (very similiar to the 

CMS More ... message). The default is nopause currently, but on

quick machines you might not be able to read the messages 

given anymore. With  option pause, the ouput will halt after

every full screen of messages for your inspection. But, anyway, those messages are also recorded  in  the preprocessor LOG-file ‘PP.LOG’ for later inspection!


-binary
generate NetRexx binary code (preceeded by the NetRexx 


option binary)  whenever possible


-strictassign
generate  NetRexx option strictassign and strict assignment 

statements.

                       -noformat
retain source line numbers (but be careful, you might no longer   

be able to read the generated NetRexx code). This option is still 

in an experimental stage (28.8.2001) and not yet fully operable..


-nowarn 
will even suppress warning messages (use with care)

-noattention
will even suppress attention messages (use with care).

-quiet

do not display messages on the screen, but do still record them in the LOG-file (file PP.LOG) 

-list

generate a detailed listing of all variables and terms used in your program. The list is written to fn.$LIST for later inspection.

As you see, the prefix ‘no’  switches an option off, to mirror the equivalent NetRexx options as far as possible.

Current Limitations:

Due to the inherent differences between Java and classic REXX, it is currently not possible to support a couple of classic REXX features, namely:

The INTERPRET statement is not available.

The SYMBOL(--) and VALUE(…) builtin functions are not available.

I do hope, that in a concerted approach, a solution for those missing links might be found later, but at the time being, I have found no way to implement those features in Java. Ideas how to overcome these restrictions are welcome, of course !

License fees:

Currently, the converter is delivered as a GNU free licence. Later, this strategy might be changed, and ideas for proper licensing techniques are welcome… It did take approximately one man-year to develop the converter and the run-time package routines, and any chances to get the (my) investments back are welcome!

System requirements:

The converter should be run on an Intel Pentium with at least 500 MHZ and 128 MB RAM (or a comparable other brand) for appropriate performance.

Support:

Technical support is available thru e-mail. In case of troubles, please send the original source file and the LOG-file of your translation effort (always available in file: PP.LOG) to my 

current e-mail address:

Thomas.Schneider@Donauland.at
Or give me a phone call at:

0043/1/81102/638      during Austrian office times

0043/1/955 57 00       during American office times

Historical Background

Note, that for historic reasons (the REXX scanner and parser routines are part of a

generalized preprocessor system) the item-classes assigned to variables do have 

different naming conventions as NetRexx currently, but the semantics are very similiar, when 

not identical at all:

Table 1: PP classes vs. NetRexx/Java primitive Types

PP
NetRexx primitive Types

Number
Byte, short, Int or Long (depending on value range). Default is Int.

Decimal
Fixed point decimal (i.e. 123.45)

Float
Floating point (i.e. 123E3)



Double
Double precision number (i.e. 1234.56D4)

Numeric
A superclass to all Numeric classes above

 

Char
A single character



Text
All Text strings (String / Rexx in NetRexx) 

Compare
The result of a comparison 

(i.e. a > b,    c = ‘this is a text ‘)

Logical
The result of a boolean expression or the constants ‘true’ and ‘false’ 

Nothing
No type (class) attached. Typically identical to void in NetRexx/Java

Any
Any possible type(class) permitted.

Short, Int, Long
A proper ‘Number’, limited by available range. 

UNumber. UShort, Uint
An ‘Unsigned Number’ (i.e. the Number must be positive or Zero)



Unumber1, …
The suffix 1 indicates, that the value 0 is NOT permitted, and the Number must be strictly positive (>=1).  

Enjoy the new vision of running classic REXX programs under JAVA!

SEITE  
4

